NAG Library Routine Document

F07HRF (ZPBTRF)

Note: before using this routine, please read the Users' Note for your implementation to check the interpretation of ***bold italicised*** terms and other implementation-dependent details.

1 Purpose

F07HRF (ZPBTRF) computes the Cholesky factorization of a complex Hermitian positive definite band matrix.

2 Specification

```
SUBROUTINE F07HRF (UPLO, N, KD, AB, LDAB, INFO)

INTEGER                 N, KD, LDAB, INFO
COMPLEX (KIND=nag_wp) AB(LDAB,*)
CHARACTER(1)            UPLO
```

The routine may be called by its LAPACK name ***zpbtrf***.

3 Description

F07HRF (ZPBTRF) forms the Cholesky factorization of a complex Hermitian positive definite band matrix A either as $A = U^{\mathrm{H}}U$ if UPLO = 'U' or $A = LL^{\mathrm{H}}$ if UPLO = 'L', where U (or L) is an upper (or lower) triangular band matrix with the same number of superdiagonals (or subdiagonals) as A.

4 References

Demmel J W (1989) On floating-point errors in Cholesky *LAPACK Working Note No. 14* University of Tennessee, Knoxville

Golub G H and Van Loan C F (1996) *Matrix Computations* (3rd Edition) Johns Hopkins University Press, Baltimore

5 Parameters

1: UPLO – CHARACTER(1) *Input*

On entry: specifies whether the upper or lower triangular part of A is stored and how A is to be factorized.

UPLO = 'U'

The upper triangular part of A is stored and A is factorized as $U^{\mathrm{H}}U$, where U is upper triangular.

UPLO = 'L'

The lower triangular part of A is stored and A is factorized as LL^{H}, where L is lower triangular.

Constraint: UPLO = 'U' or 'L'.

2: N – INTEGER *Input*

On entry: n, the order of the matrix A.

Constraint: N $\geq$ 0.

3: KD – INTEGER *Input*

On entry: k_d, the number of superdiagonals or subdiagonals of the matrix A.

Constraint: KD ≥ 0.

4: AB(LDAB,$*$) – COMPLEX (KIND=nag_wp) array *Input/Output*

Note: the second dimension of the array AB must be at least $\max(1, \mathrm{N})$.

On entry: the n by n Hermitian positive definite band matrix A.

The matrix is stored in rows 1 to $k_d + 1$, more precisely,

if UPLO = 'U', the elements of the upper triangle of A within the band must be stored with element A_{ij} in AB$(k_d + 1 + i - j, j)$ for $\max(1, j - k_d) \leq i \leq j$;

if UPLO = 'L', the elements of the lower triangle of A within the band must be stored with element A_{ij} in AB$(1 + i - j, j)$ for $j \leq i \leq \min(n, j + k_d)$.

On exit: the upper or lower triangle of A is overwritten by the Cholesky factor U or L as specified by UPLO, using the same storage format as described above.

5: LDAB – INTEGER *Input*

On entry: the first dimension of the array AB as declared in the (sub)program from which F07HRF (ZPBTRF) is called.

Constraint: LDAB $\geq$ KD $+ 1$.

6: INFO – INTEGER *Output*

On exit: INFO $= 0$ unless the routine detects an error (see Section 6).

6 Error Indicators and Warnings

Errors or warnings detected by the routine:

INFO < 0

If INFO $= -i$, the ith parameter had an illegal value. An explanatory message is output, and execution of the program is terminated.

INFO > 0

If INFO $= i$, the leading minor of order i is not positive definite and the factorization could not be completed. Hence A itself is not positive definite. This may indicate an error in forming the matrix A. There is no routine specifically designed to factorize a band matrix which is not positive definite; the matrix must be treated either as a nonsymmetric band matrix, by calling F07BRF (ZGBTRF) or as a full matrix, by calling F07MRF (ZHETRF).

7 Accuracy

If UPLO = 'U', the computed factor U is the exact factor of a perturbed matrix $A + E$, where

$$|E| \leq c(k+1)\epsilon \left|U^{\mathrm{H}}\right||U|,$$

$c(k+1)$ is a modest linear function of $k+1$, and ϵ is the ***machine precision***.

If UPLO = 'L', a similar statement holds for the computed factor L. It follows that $\left|e_{ij}\right| \leq c(k+1)\epsilon\sqrt{a_{ii}a_{jj}}$.

8 Further Comments

The total number of real floating point operations is approximately $4n(k+1)^2$, assuming $n \gg k$.

A call to F07HRF (ZPBTRF) may be followed by calls to the routines:

F07HSF (ZPBTRS) to solve $AX = B$;

F07HUF (ZPBCON) to estimate the condition number of A.

The real analogue of this routine is F07HDF (DPBTRF).

9 Example

This example computes the Cholesky factorization of the matrix A, where

$$A = \begin{pmatrix} 9.39 + 0.00i & 1.08 - 1.73i & 0.00 + 0.00i & 0.00 + 0.00i \\ 1.08 + 1.73i & 1.69 + 0.00i & -0.04 + 0.29i & 0.00 + 0.00i \\ 0.00 + 0.00i & -0.04 - 0.29i & 2.65 + 0.00i & -0.33 + 2.24i \\ 0.00 + 0.00i & 0.00 + 0.00i & -0.33 - 2.24i & 2.17 + 0.00i \end{pmatrix}.$$

9.1 Program Text

```
    Program f07hrfe

!     F07HRF Example Program Text

!     Mark 24 Release. NAG Copyright 2012.

!     .. Use Statements ..
      Use nag_library, Only: nag_wp, x04dff, zpbtrf
!     .. Implicit None Statement ..
      Implicit None
!     .. Parameters ..
      Integer, Parameter               :: nin = 5, nout = 6
!     .. Local Scalars ..
      Integer                          :: i, ifail, info, j, kd, ldab, n
      Character (1)                    :: uplo
!     .. Local Arrays ..
      Complex (Kind=nag_wp), Allocatable :: ab(:,:)
      Character (1)                    :: clabs(1), rlabs(1)
!     .. Intrinsic Procedures ..
      Intrinsic                        :: max, min
!     .. Executable Statements ..
      Write (nout,*) 'F07HRF Example Program Results'
!     Skip heading in data file
      Read (nin,*)
      Read (nin,*) n, kd
      ldab = kd + 1
      Allocate (ab(ldab,n))

!     Read A from data file

      Read (nin,*) uplo
      If (uplo=='U') Then
        Do i = 1, n
          Read (nin,*)(ab(kd+1+i-j,j),j=i,min(n,i+kd))
        End Do
      Else If (uplo=='L') Then
        Do i = 1, n
          Read (nin,*)(ab(1+i-j,j),j=max(1,i-kd),i)
        End Do
      End If

!     Factorize A
!     The NAG name equivalent of zpbtrf is f07hrf
      Call zpbtrf(uplo,n,kd,ab,ldab,info)

      Write (nout,*)
      Flush (nout)
      If (info==0) Then

!       Print factor
```

```
!       ifail: behaviour on error exit
!              =0 for hard exit, =1 for quiet-soft, =-1 for noisy-soft
        ifail = 0
        If (uplo=='U') Then

          Call x04dff(n,n,0,kd,ab,ldab,'Bracketed','F7.4','Factor','Integer', &
            rlabs,'Integer',clabs,80,0,ifail)

        Else If (uplo=='L') Then

          Call x04dff(n,n,kd,0,ab,ldab,'Bracketed','F7.4','Factor','Integer', &
            rlabs,'Integer',clabs,80,0,ifail)

        End If

      Else
        Write (nout,*) 'A is not positive definite'
      End If

    End Program f07hrfe
```

9.2 Program Data

```
F07HRF Example Program Data
  4  1                                                    :Values of N and KD
  'L'                                                     :Value of UPLO
 ( 9.39, 0.00)
 ( 1.08, 1.73) ( 1.69, 0.00)
               (-0.04,-0.29) ( 2.65, 0.00)
                             (-0.33,-2.24) ( 2.17, 0.00)  :End of matrix A
```

9.3 Program Results

```
 F07HRF Example Program Results

 Factor
                     1                2                3                4
 1  ( 3.0643, 0.0000)
 2  ( 0.3524, 0.5646) ( 1.1167, 0.0000)
 3                    (-0.0358,-0.2597) ( 1.6066, 0.0000)
 4                                      (-0.2054,-1.3942) ( 0.4289, 0.0000)
```